
 07
Faça o que eu fiz na aula

Vamos criar uma view, para isso vamos navegar para a transação SE11. Então vamos selecionar a opção Data type e, como

sugestão de nome, vamos inserir ZSPFLI_ATRASOS_ESTRUTURA . Após isso podemos clicar no botão Create.

Após clicar no botão Create um popup será apresentado perguntando que Data type desejamos criar. Vamos selecionar a

opção Structure e então podemos con�rmar.

Vamos criar uma estrutura que possua campos da tabela SPFLI e da tabela ZSPFLI_ATRASOS.

Durante a ativação da nossa view recebemos uma mensagem informando que a nossa estrutura foi ativada, entretanto, com

avisos. Vamos então veri�car quais são os avisos que necessitamos tratar.

Novamente temos avisos referente ao Enhancement Category. Vamos acessar acessar a barra de menu e navegar em Extras >

Enhancement Category.

Um popup será exibido, e então selecionamos a opção Cannot Be Enhanced.

Após a con�guração feita, podemos ativar novamente a nossa view e então teremos apenas a mensagem de sucesso no

rodapé, informando que o objeto foi ativado com sucesso.

Com a view criada vamos voltar ao nosso código e alterar o nosso select. Para isso, vamos atualizar o nosso código com o

seguinte:

Código completo

 
PROGRAM z_controle_de_voos. 
 
INCLUDE z_controle_de_voos_declaracoes. 
INCLUDE z_controle_de_voos_selecao. 
INCLUDE z_controle_de_voos_pbo. 
INCLUDE z_controle_de_voos_pai. 
 
*&---------------------------------------------------------------------*
*& Include          Z_CONTROLE_DE_VOOS_DECLARACOES 
*&---------------------------------------------------------------------* 
 
TABLES zspfli_atrasos . 
 
DATA spfli TYPE spfli . 
 
DATA: custom_container TYPE REF TO cl_gui_custom_container, 
      alv              TYPE REF TO cl_gui_alv_grid, 
      fieldcat         TYPE lvc_t_fcat. 
 
DATA relatorio_atrasos TYPE STANDARD TABLE OF zspfli_atrasos_estrutura . 
 
*&---------------------------------------------------------------------*
*& Include          Z_CONTROLE_DE_VOOS_SELECAO 
*&---------------------------------------------------------------------* 
 
 
SELECTION-SCREEN: BEGIN OF SCREEN 9999 AS SUBSCREEN . 
SELECTION-SCREEN: BEGIN OF BLOCK bloco1 WITH FRAME TITLE TEXT-001 . 
 
SELECT-OPTIONS s_carrid FOR spfli-carrid NO INTERVALS. 
SELECT-OPTIONS s_connid FOR spfli-connid NO INTERVALS. 
SELECT-OPTIONS s_cityfr FOR spfli-cityfrom NO INTERVALS. 
SELECT-OPTIONS s_cityto FOR spfli-cityto NO INTERVALS. 

SELECTION-SCREEN: END OF BLOCK bloco1. 
SELECTION-SCREEN: END OF SCREEN 9999 . 
 
 
*&---------------------------------------------------------------------*
*& Include          Z_CONTROLE_DE_VOOS_PBO 
*&---------------------------------------------------------------------* 
 
 
MODULE pbo OUTPUT. 
 
  DATA coluna LIKE LINE OF fieldcat . 
 
  SET PF-STATUS '1000'. 
 
  IF custom_container IS NOT BOUND . 
 
    CREATE OBJECT custom_container 
      EXPORTING 
        container_name = 'CC_ALV'. 
 
    CREATE OBJECT alv 
      EXPORTING 
        i_parent = custom_container. 
 
    CALL FUNCTION 'LVC_FIELDCATALOG_MERGE' 
      EXPORTING 
        i_structure_name = 'ZSPFLI_ATRASOS_ESTRUTURA' 
      CHANGING 
        ct_fieldcat      = fieldcat. 
 
 
    CALL METHOD alv->set_table_for_first_display 
      CHANGING 
        it_outtab       = relatorio_atrasos    " Output Table 
        it_fieldcatalog = fieldcat.     " Field Catalog 
 
  ENDIF . 
 
  CALL METHOD alv->refresh_table_display( ) . 
 
ENDMODULE. 
 
 
*&---------------------------------------------------------------------*
*&  Include           Z_CONTROLE_DE_VOOS_PAI 
*&---------------------------------------------------------------------* 
 
MODULE pai INPUT. 
 
  CASE sy-ucomm. 
 
    WHEN 'VOLTAR' OR 'CANCELAR' OR 'SAIR'. 
 
      LEAVE PROGRAM. 
 
    WHEN 'EXEC'. 

      IF zspfli_atrasos-dia_do_voo IS INITIAL . 
 
        SELECT carrid, connid, countryfr, 
                      cityfrom, airpfrom, countryto, 
                        cityto, airpto, spfli~fltime, deptime, 
                          arrtime 
               FROM spfli 
               INTO TABLE @relatorio_atrasos 
               WHERE carrid IN @s_carrid 
                 AND connid IN @s_connid 
                 AND cityfrom IN @s_cityfr 
                 AND cityto IN @s_cityto. 
 
      ELSE . 
 
        SELECT spfli~carrid, spfli~connid, spfli~countryfr, 
                         spfli~cityfrom, spfli~airpfrom, spfli~countryto, 
                           spfli~cityto, spfli~airpto, spfli~fltime, spfli~deptime, 
                             spfli~arrtime, zspfli_atrasos~dia_do_voo, zspfli_atrasos~atraso 
                   FROM spfli AS spfli 
                   INNER JOIN zspfli_atrasos AS zspfli_atrasos 
                   ON spfli~carrid = zspfli_atrasos~carrid 
                     AND spfli~connid = zspfli_atrasos~connid 
                   INTO TABLE @relatorio_atrasos 
                    WHERE spfli~carrid IN @s_carrid 
                          AND spfli~connid IN @s_connid 
                          AND spfli~cityfrom IN @s_cityfr 
                          AND spfli~cityto IN @s_cityto 
                          AND zspfli_atrasos~dia_do_voo = @zspfli_atrasos-dia_do_voo. 
 
        IF sy-subrc = 0 . 
 
          DATA wa_relatorio_atrasos LIKE LINE OF relatorio_atrasos . 
 
          LOOP AT relatorio_atrasos INTO wa_relatorio_atrasos . 
 
            IF wa_relatorio_atrasos-atraso IS NOT INITIAL . 
 
              wa_relatorio_atrasos-arrtime = wa_relatorio_atrasos-arrtime + wa_relatorio_atrasos-atr
              MODIFY relatorio_atrasos FROM  wa_relatorio_atrasos . 
 
            ENDIF . 
 
          ENDLOOP . 
        ENDIF . 
      ENDIF . 
 
 
    WHEN 'BTN_ATRASO'. 
 
      DATA: wa_spfli       LIKE LINE OF relatorio_atrasos, 
            wa_spfliatraso TYPE zspfli_atrasos. 
 
      LOOP AT relatorio_atrasos INTO wa_spfli . 
 
        wa_spfliatraso-carrid = wa_spfli-carrid . 
        wa_spfliatraso-connid = wa_spfli-connid . 

        wa_spfliatraso-dia_do_voo = zspfli_atrasos-dia_do_voo. 
        wa_spfliatraso-atraso = zspfli_atrasos-atraso . 
 
        INSERT zspfli_atrasos FROM wa_spfliatraso . 
 
        IF sy-subrc = 0 . 
 
          MESSAGE 'Atraso registrado com sucesso'  TYPE 'S' . 
 
        ELSE . 
 
          MESSAGE 'Erro ao registrar atraso no banco de dados'  TYPE 'S' DISPLAY LIKE 'E' . 
 
        ENDIF . 
 
      ENDLOOP . 
 
  ENDCASE. 
 
ENDMODULE.


